

# Schnell und schön – QUICKSORT

(von C.A.R. Hoare 1962)



( Am Beispiel einer Artikelliste )

```
procedure CListe.quicksort (Anfang, Ende : integer);  
  
var links, rechts : integer;  
    Vergleichswert : Attributtyp_nach_dem_sortiert_wird;  
  
begin  
    links := Anfang;  
    rechts := Ende;  
    Vergleichswert := Items[(links + rechts) DIV 2].getArtNr;  
  
    repeat  
  
        while (Items[links].getArtNr < Vergleichswert) do inc(links);  
        while (Items[rechts].getArtNr > Vergleichswert) do dec(rechts);  
  
        if (links <= rechts)  
        then begin  
            Swap(Items[links], Items[rechts]);  
            inc(links);  
            dec(rechts);  
        end; (* i f *)  
  
    until (rechts < links);  
  
    if (rechts > Anfang)  
    then quicksort(Anfang, rechts);  
  
    if (links < Ende)  
    then quicksort(links, Ende);  
  
end; (* q u i c k s o r t *)
```

Linken und rechten „Zeiger“ initialisieren und ein möglichst günstiges Vergleichselement bestimmen.

Problem: Wo ist der Median?

Lösung:

Zunächst von links her ‘nen „Falschen“ suchen.

Danach von rechts her ‘nen „Falschen“ suchen.

Es darf natürlich nur getauscht werden, wenn die beiden „Zeiger“ beim „Falschen suchen“ nicht aneinander vorbeigerauscht sind.  
Die Einbeziehung von ‘=’ ist wg. des singulären Falles „zufällig Median getroffen“ wichtig.

Sind die beiden „Zeiger“ aneinander vorbeigerauscht, ist der Teilbereich von „Anfang“ bis „Ende“ mindestens grob sortiert. Der Rest wird rekursiv erledigt.

Wenn auf der linken Seite noch wenigstens 2 Elemente vorhanden sind.  
(Beachte: Der rechte „Zeiger“ ist gegen den „Anfang“ gewandert!)

Wenn auf der rechten Seite noch wenigstens 2 Elemente vorhanden sind.  
(Beachte: Der linke „Zeiger“ ist gegen das „Ende“ gewandert!)